

Building Intelligent Macros: Driving a Variable Parameter System with Metadata

Arthur L. Carpenter, California Occidental Consultants, Anchorage, Alaska

ABSTRACT

When faced with generating a series of reports, graphs, and charts; we will often use the macro language to simplify the process. Commonly we will write a series of generalized macros, each with the capability of creating a variety of outputs that depend the macro parameter inputs. For large projects, potentially with hundreds of outputs, controlling the macro calls can itself become difficult.

The use of control files (metadata) to organize and process when a single macro is to be executed multiple times was discussed by Rosenbloom and Carpenter (2015). But those techniques only partially help us when multiple macros, each with its own set of parameters, are to be called. This paper discusses a technique that allows you to control the order of macro calls along with each macro's parameter set, while using a metadata control file.

KEYWORDS

Metadata, macro language, CALL EXECUTE

INTRODUCTION

When a single macro is to be called multiple times with varying parameters, a metadata control file such as the one discussed in detail in Rosenbloom and Carpenter (2015) can be used. Typically this metadata file will contain one row per macro call and one column per parameter. Depending on the ultimate objective of the programmer, the metadata is used to generate a series of macro variable lists or a series of CALL EXECUTES.

Although this approach works great for multiple calls to a single macro, it is not very efficient when multiple macros are to be called. And it especially does not work when the different macros have very different parameter structures. In this situation there may still be the need for a metadata control file, however a different approach must be taken.

The basic example used here deals with the generation of an Annual Report. The report is composed of a series of tables, charts, and graphs. Generalized macros have been written, which when executed create one or more of the components for the final report. Because the macros have been generalized, each macro can actually create any number of similar tables by varying the macro's parameters. This allows us to generate hundreds of different tables with only a few macros.

GENERALIZED MACROS

For the purposes of this paper the use of three simple macros will be demonstrated. The beauty of the technique that is to be described is that it is completely expandable to any number of macros, each with any number of parameters.

```
%macro printit(dsn=,varlist=,title=);  
title1 "&title";  
proc print data=&dsn;  
  var &varlist;  
  run;  
%mend printit;  
  
%macro chart(dsn=, gvar=, yvar=, type= ,title=);  
title1 "&title";  
proc gchart data=&dsn;  
  vbar &yvar/group=&gvar type=&type;  
  run;  
  quit;  
%mend chart;  
  
%macro summarize(dsn=,classlist=, var=);  
proc summary data=&dsn;  
  class &classlist;  
  var &var;  
  output out=summry mean= n= stderr=/autoname;  
  run;  
%mend summarize;
```

The %PRINTIT macro generates a simple data listing. It has three parameters, which are used to name the data set, the variables to print, and a title.

The %CHART macro uses the GCHART procedure to generate a vertical bar chart. Parameters are used to name the data set of interest, the vertical variable, a grouping variable, the type of the chart to create, and a title.

The %SUMMARIZE macro calls a PROC SUMMARY step and is used to summarize data prior to the generation of a report. The data set generated by this procedure can in turn be used by the %PRINTIT macro.

Figure 1

The macro calls that generate our silly little annual report be called in the order that we would like our report tables to appear.

```
%printit(dsn=sashelp.class,varlist=name sex height weight,title=class data)  
%chart(dsn=sashelp.class,yvar=weight,gvar=sex,type=percent, title=Weight distribution  
for gender)  
%summarize(dsn=sashelp.class,classlist=sex age,var=height weight)  
%printit(dsn=summry,varlist=_all_,title=Summarized class data)
```

Figure 2

Clearly with a number of macros being called and a series of often different parameters, the calls themselves are not easily read. The construction of this type of series of macro calls is prone to error. It is easy to miss specify a parameter, skip a macro call, place calls in the wrong order, and other such coding problems. One way to make the code easier to read and maintain is to place one parameter per line of code.

```

%printit(dsn=sashelp.class,
         varlist=name sex height weight,
         title=class data)
%chart(dsn=sashelp.class,
       yvar=weight,
       gvar=sex,
       type=percent,
       title=Weight distribution for gender)
%summarize(dsn=sashelp.class,
           classlist=sex age,
           var=height weight)
%printit(dsn=summry,
         varlist=_all_,
         title=Summarized class data)
    
```

Figure 3 shows the same macro calls as appear in Figure 2, however it is now much easier to visualize the individual parameters in each macro call. It is also much easier to assess the order of the macro calls and whether or not that order of the macro calls is appropriate for our annual report. All this and we have not yet even added any comments!

Figure 3

BUILDING METADATA

Creating a series of calls to generalized macros is often sufficient for most applications, but when you have hundreds of macro calls or if you calling macros that have more than three or four parameters, even coding them as is done in Figure 3 is problematic. One approach that can be used to simplify the control of large numbers of macro calls is through the use of metadata.

The metadata approach moves all of the critical information provided in the code into data, while leaving the syntax

	A	B	C
1	macro	parm	value
2	printit	dsn	sashelp.class
3	printit	varlist	name sex height weight
4	printit	title	class data
5	chart	dsn	sashelp.class
6	chart	yvar	weight
7	chart	gvar	sex
8	chart	type	percent
9	chart	title	Weight Distribution for Gender
10	summarize	dsn	sashelp.class
11	summarize	classlist	sex age
12	summarize	var	height weight
13	printit	dsn	summry
14	printit	varlist	_all_
15	printit	title	Summarized Class Data

Figure 4

behind. The information in Figure 4 is the same as in Figure 3 except all the code has been removed. In this particular example the metadata is being stored in a XLS file, however it could just as easily been in a CSV file, TEXT file, or any other file form that can be imported into SAS.

One immediate advantage of this approach is that the metadata can be maintained by someone with little or no SAS knowledge. As we will see as we look at how this metadata is used, the technique is highly expandable, and can support drastic changes to the metadata without causing any changes to the code that uses the metadata.

In the approach used here we will read this metadata file into SAS using a PROC IMPORT step. Generally this will be the easiest way to bring the metadata into SAS, but as long as we end up with a SAS data set, it does not matter.

The PROC IMPORT is completely straightforward. In the PROC IMPORT step shown in Figure 5, the XLS file

```
proc import file="&somepath\2_MetaData.xls"
            out=metadata
            replace
            dbms=excelcs;

run;
```

2_METADATA.XLS is imported and the SAS data set WORK.METADATA is created. This data set has the same information in it as was contained in the XLS file. Inspection of the data set WORK.METADATA, shown in Figure 6

Figure 5

We now have the ability to take advantage of the metadata in a DATA step. We can build lists of macro variables, as was demonstrated in Rosenbloom and Carpenter (2015), or we can build the macro calls directly and execute them through the use of the CALL EXECUTE routine. Because in this example we are only interested in executing a series of macro calls, the CALL EXECUTE approach will be used.

	macro	parm	value
1	printit	dsn	sashelp.class
2	printit	varlist	name sex height weight
3	printit	title	class data
4	chart	dsn	sashelp.class
5	chart	yvar	weight
6	chart	gvar	sex
7	chart	type	percent
8	chart	title	Weight Distribution for Gender
9	summarize	dsn	sashelp.class
10	summarize	classlist	sex age
11	summarize	var	height weight
12	printit	dsn	summy
13	printit	varlist	_all_
14	printit	title	Summarized Class Data

Figure 6

BUILDING AND EXECUTING THE MACRO CALLS

Except for the actual SAS syntax the metadata contains all the information that we need to construct the macro calls. Our approach will be to read the data and use the power of the DATA step to create the macro call including its parameters. The macro calls will then be submitted for execution.

For the first macro call, a call to the %PRINTIT macro, we need to construct the code shown in Figure 7. Because this

```
%printit(dsn=sashelp.class,
          varlist=name sex height weight,
          title=class data)
```

code will be submitted for execution using CALL EXECUTE, the macro call itself must be masked during DATA step execution. This allows the CALL EXECUTE routine to add the macro call to a buffer for execution after the termination of the DATA step.

Figure 7

We can mask the macro call through the use of the %NRSTR macro quoting function. Effectively we need to construct

```
call execute('%nrstr(%printit(dsn=sashelp.class,
                             varlist=name sex height weight,
                             title=class data))');
```

an argument for CALL EXECUTE that both contains and masks the macro call. The %NRSTR function prevents the DATA step from seeing the

Figure 8

%PRINTIT as a macro call, and the macro call is passed out of the DATA step, where it is executed after the DATA step completes its execution.

The DATA step in Figure 9 will read the metadata and utilize the information that it contains to build the masked macro call.

```

data _null_;
  length string $500; ❶
  retain string;
  set metadata;
  by macro notsorted; ❷
  if first.macro then do; ❸
    string = cats('%nrstr%',macro, '(',parm, '=',value);
  end;
  else string = cats(string, ',',parm, '=',value); ❹
  if last.macro then do; ❺
    string = cats(string, '))');
    put string=; ❻
    call execute(string); ❼
  end;

```

Figure 9

❶ A character string variable is defined that will hold the macro call. This particular instance is limited to 500 characters.

❷ The BY statement enables the use of FIRST. and LAST. processing. The NOTSORTED keyword is needed because although the incoming data is grouped by macro name, it is not sorted by macro name. Note this logic fails if there are two successive calls to the same macro (see

Extension 2 below for a coding solution).

- ❸ This is the first observation for this macro call add the %NRSTR and the macro name with its % sign along with the first parameter to the variable STRING.
- ❹ This is not the first observation for this macro, add the parameter and its value to the growing list.
- ❺ This is the last observation for this macro, add the last two closing parentheses.
- ❻ The value of STRING is written to the SAS Log so that you can see what it contains. This is of course optional and is only shown here to demonstrate the value held by the variable STRING.
- ❼ The value of the variable STRING, which contains the masked macro call, is passed to the buffer for execution after the DATA step terminates.

The SAS Log in Figure 10 shows the results of the PUT statement ❻, which was included in the DATA step so that we could visualize the generated code.

```

string=%nrstr(%printit(dsn=sashelp.class,varlist=name sex height weight,title=class data))
string=%nrstr(%chart(dsn=sashelp.class,yvar=weight,gvar=sex,type=percent,title=Weight
Distribution for Gender))
string=%nrstr(%summarize(dsn=sashelp.class,classlist=sex age,var=height weight))
string=%nrstr(%printit(dsn=summry,varlist=_all_,title=Summarized Class Data))

```

Figure 10

The SAS Log also shows us the code that is actually executed after being written to the buffer by CALL EXECUTE. The first %PRINTIT macro call is shown in Figure 11.

NOTE: CALL EXECUTE generated line.

```
1 + %printit(dsn=sashelp.class,varlist=name sex height weight,title=class data)
```

Figure 11

EXTENSION 1: MACRO CALLS WITHOUT PARAMETERS

The DATA step in Figure 9 assumes that each macro has at least one parameter. While this will generally be true in it will not always be true. Fortunately the coding changes are simple and fairly straightforward. We merely need to detect macro call without parameters.

```
data _null_;
  length string $500;
  retain string;
  set metadata;
  by macro notsorted;
  if first.macro then do;
    if parm= ' ' then do; ❶
      string = cats('%nrstr%',macro); ❷
    end;
    else do;
      string = cats('%nrstr%',macro, '(',parm, '=',value);
    end;
  end;
  else string = cats(string,',',parm, '=',value);
  if last.macro then do;
    if parm= ' ' then do; ❸
      string = cats(string,')');
    end;
    else do;
      string = cats(string,')')');
    end;
  put string=;
  call execute(string);
end;
run;
```

Figure 12

This is actually simple if we assume that the PARM variable will only be missing for macros without parameters. As this seems to be a reasonable assumption we can add the necessary logic to the DATA step in Figure 9.

❶ The assumption is that the PARM variable will be missing for macros without parameters.

❷ Only the %NRSTR and macro name need to be concatenated.

❸ Only one closing parenthesis is needed (to close the %NRSTR function).

EXTENSION 2: CONTROLLING CALL ORDER

The DATA steps in Figures 9 and 12 will both fail if the same macro is called successively. This is an artifact of the use of the macro name in the BY statement. As a result there is no way to differentiate between two different calls to the same macro when one directly follows the other. A common way to solve this problem is to add a numbering system of some kind to the metadata. For reporting systems very often a table number will be included anyway and we can take advantage of the table number.

	A	B	C	D
1	table	macro	parm	value
2	1.1.1	chart	dsn	sashelp.class
3	1.1.1	chart	yvar	weight
4	1.1.1	chart	gvar	sex
5	1.1.1	chart	type	percent
6	1.1.1	chart	title	Table: Weight Distribution fo
7	1.1.2	summarize	dsn	sashelp.class
8	1.1.2	summarize	classlist	sex age
9	1.1.2	summarize	var	height weight
10	1.1.2	printit	dsn	summry
11	1.1.2	printit	varlist	_all_
12	1.1.2	printit	title	Table: Summarized Class Dat
13	1.1.3	printit	dsn	sashelp.class
14	1.1.3	printit	varlist	name sex height weight
15	1.1.3	printit	title	Table: class data

Figure 13

Adding a table number to the metadata gives us a way to not only augment our titles, but to solve the problem of successive macro calls to the same macro. In Figure 13 the TABLE column has been added and the order of the macro calls has been changed. Notice that now that there are two successive calls to %PRINTIT, but that each has its own distinct table number.

In the titles 'Table:' has been added. In the DATA step that processes the metadata, the table number will be added to the title using this code. The table number is added using the TRANWRD function.

The remainder of the DATA step remains unchanged from Figure 12.

```

data _null_;
  length string $500;
  retain string;
  set metadata;
  by table macro notsorted; ❶

  * Insert Table number if requested; ❷
  value = tranwrđ(value, 'Table:', catx(' ', 'Table:', table));

  ... Code not shown ...

```

Figure 14

❶ The variable TABLE has been added to the BY statement. This ensures that successive macro calls of the same macro can be distinguished.

❷ The TRANWRD function is used to add the table number wherever the 'Table:' is detected.

The remainder of the DATA step remains unchanged from Figure 12.

EXTENSION 3: WORKING WITH LISTS WITHIN A PARAMETER

In the previous examples the variable list (PARAM=VARLIST) for the %PRINTIT macro were on a single line. When the user does not have full control over how the metadata control table is created, it may not be possible to place all the items in a list on a single line. When this can happen our data step needs to be flexible enough to accommodate this form of the metadata.

A portion some metadata that is in this form is shown in Figure 15. Rather than appearing on a single line the list of variables used with the VARLIST parameters appears with one variable per line. One obvious advantage of this form is that the list of items can be very long.

15	1.1.3	printit	dsn	sashelp.class
16	1.1.3	printit	varlist	name
17	1.1.3	printit	varlist	sex
18	1.1.3	printit	varlist	height
19	1.1.3	printit	varlist	weight
20	1.1.3	printit	title	Table: class data

Figure 15

The changes to the DATA step require us to detect lists that span multiple observations. This is accomplished by adding the variable PARM to the BY statement, which gives us the ability to use FIRST. processing on the PARM variable.

```
data _null_;
  length string $500;
  retain string;
  set metadata;
  by table macro parm notsorted; ❶

  value = tranwrd(value,'Table:',catx(' ','Table:',table));

  if first.macro then do;
    string = cats('%nrstr%',macro); ❷
    if parm ne ' ' then string = cats(string,',' ,parm,'=',value);
  end;
  else if first.parm then string = cats(string,',' ,parm,'=',value); ❸
  else string = catx(' ',string,value); ❹
  if last.macro then do;
    string = cats(string,')'); ❺
    if parm ne ' ' then string = cats(string,')');
    put string=;
    call execute(string);
  end;
run;
```

Figure 16

parentheses was simplified in this version of the DATA _NULL_ step.

- ❶ The variable PARM is added to the BY statement.
- ❷ The logic to detect macro calls without parameters has been simplified.
- ❸ If this is the first occurrence of this parameter, add the parameter name and equal sign as well as its value to the macro call.
- ❹ If it is not the first occurrence of the parameter then just add the parameter value.
- ❺ The logic to write the closing

CONCLUSION


Applications for the use of metadata control files to manage the execution of a series of macro calls are many and varied. Most commonly the metadata is of a form where each observation of the metadata forms a distinct macro call. While this form of metadata may be the most commonly encountered, it is by no means the only metadata form. Through the power of the DATA step we make use of metadata that may come to us in a variety of forms.

In this paper the discussion centers on metadata in a vertical format. This type of control file tends to use multiple observations to describe each macro call and may require the use of DATA step logic to build each macro call. But because we are working in the DATA step, this approach can be highly flexible.

REFERENCES

Carpenter, Art, 2016, *Carpenter's Complete Guide to the SAS® Macro Language, Third Edition*, SAS Institute Inc, Cary, NC. <http://support.sas.com/publishing/authors/carpenter.html>

Fehd, Ronald and Art Carpenter, 2007, "List Processing Basics: Creating and Using Lists of Macro Variables" by Ronald Fehd and Art Carpenter which was presented at the 2007 SAS Global Forum (Paper 113-2007). The discussion of the paper looks at different approaches used in the automation of programs by using various kinds of macro variable lists. This paper appears in proceedings of a number of conferences, including: SASGF(2007), WUSS (2008), MWSUG (2009), SESUG (2009). <http://www.caloxy.com/papers/72Lists.pdf>



Rosenbloom, Mary F. O. and Carpenter, Arthur L., 2015, "Are You a Control Freak? Control Your Programs – Don't Let Them Control You!", presented at the SAS Global Forum 2015 Conference in Dallas, Texas (paper 2220-2015). <http://support.sas.com/resources/papers/proceedings15/2220-2015.pdf>

ABOUT THE AUTHOR

Art Carpenter is a SAS Certified Advanced Professional Programmer and his publications list includes; five books and numerous papers and posters presented at SAS Global Forum, SUGI, PharmaSUG, WUSS, and other regional conferences. Art has been using SAS since 1977 and has served in various leadership positions in local, regional, and international user groups.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the authors at:

Arthur L. Carpenter
California Occidental Consultants
10606 Ketch Circle
Anchorage, AK 99515

(907) 865-9167
art@caloxy.com
www.caloxy.com



View Art Carpenter's paper presentations page at:
http://www.sascommunity.org/wiki/Presentations:ArtCarpenter_Papers_and_Presentations

TRADEMARK REFERENCES

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.